
Cut

In general, if we have:

foo :- a, b, c, !, d, e, f.

foo :- g, h.

• Prolog backtracks among goals a, b, and c.

• Prolog backtracks among goals d, e, and f.

• If d fails, Prolog will not try to re-satisfy c.

• Prolog will not try to re-satisfy the parent goal foo.

If Prolog gets to the Cut, trying to satisfy foo, the only way to
achieve this would be to satisfy d, e, f.

Cut

The goal “!”, pronounced “cut”, always succeeds immediately.
It has an important side effect: Once it is satisfied, it disallows
either:

• backtracking back over the cut, or

• backtracking and applying a different clause of the parent
predicate to satisfy the present goal.

Cut

% add(X,L1,L2) holds iff L2 is the result of adding X

% to L1, such that no duplicates are added to the list.

add(X,L,L) :- member(X,L).

add(X,L,[X|L]).

What is the result of the following call to add? Why?

?- add(a, [a, b, c], L).

L = [a, b, c] ;

L = [a, a, b, c].

Cut

Here’s how to do it with cut:

addC(X,L,L) :- member(X,L), !.

addC(X,L,[X|L]).

Think about why this works.

Cut

Recall our add and addC predicates:

add(X,L,L) :- member(X,L).

add(X,L,[X|L]).

addC(X,L,L) :- member(X,L), !.

addC(X,L,[X|L]).

?- add(X, [a,b], [a,b]).

?- addC(X, [a,b], [a,b]).

Cut: notes

add(X,L,L) :- member(X,L).

add(X,L,[X|L]).

addC(X,L,L) :- member(X,L), !.

addC(X,L,[X|L]).

?- add(X, [a,b], [a,b]).

X = a ;

X = b ;

false.

?- addC(X, [a,b], [a,b]).

X = a.

Cut: notes

?- add(a, L1, [a,b]).

L1 = [a, b] ;

L1 = [b].

?- addC(a, L1, [a,b]).

L1 = [a, b].

Notes:
• We wrote addC to work when L2 is the only uninstantiated

variable.

• Cut may work only for restricted uses of a predicate.

• Need to clearly define your preconditions & assumptions.

Cut and If-then-else

One common use of Cut is to mimic the “if-then-else” construct
found in imperative languages.

In Prolog, If p(X) then q(X), else r(X) can be written as follows:

s(X) :- p(X) -> q(X) ; r(X).

Here’s an example:

max(X,Y,Z) :- X =< Y -> Z=Y ; Z=X.

Cut

Cut can be used to improve the efficiency of search by reducing the
search space, i.e., by pruning the search tree.

For example, when two predicates are mutually exclusive:

q(X) :- even(X), a(X).

q(X) :- odd(X), b(X).

With cut

q(X) :- even(X), !, a(X).

q(X) :- odd(X), b(X).

Cut

Cut can remove unwanted answers. Consider the Family Database:

1. male(charlie). 6. parent(charlie,bob).

2. male(albert). 7. parent(eve,bob).

3. male(bob). 8. parent(charlie,albert).

9. parent(eve,albert).

4. female(eve). 10. parent(charlie,alice).

5. female(alice). 11. parent(eve,alice).

% son(?X) iff X is a son

son(X) :- parent(_,X), male(X).

?- son(X).

Cut

We want to rewrite the rule son so that it does not generate
duplicate answers.

Cut

We want to rewrite the rule son so that it does not generate
duplicate answers.

son(X) :- parent(_,X), male(X), !.

?- son(X).

X = bob.

What about albert? Think:

• Any male is potentially a good answer, so we want to try all
of them: can’t put “cut” after male in the same rule.

• If a male has 2 parents, we only want to list him once as the
answer: want to put “cut” after parent.

Cut

Try again:

son(X) :- parent(_,X), !, male(X).

?- son(X).

X = bob.

hmmm, ... what is happening here?

Cut

Try again:

son(X) :- parent(_,X), !, male(X).

?- son(X).

X = bob.

hmmm, ... what is happening here?

• Prolog does not re-satisfy parent( ,X) ⇒ desired effect of
using Cut.

• but Prolog also does not re-satisfy son(X) ⇒ an undesired /
unwanted effect of Cut.

Solution?

Cut

Think again:

• Any male is potentially a good answer, so we want to try all
of them: can’t put “cut” after male in the same rule.

• If a male has 2 parents, we only want to list him once as the
answer: want to put “cut” after parent.

• We want to find all values of X that satisfy son(X), so can’t
put “cut” inside the rule for son.

16



Cut

Result:

son(X) :- male(X), hasparent(X).

hasparent(X) :- parent(_,X), !.

?- son(X).

X = bob ;

X = albert.

Cut

What about sibling?

sibling(X,Y) :- parent(P,X), parent(P,Y).

?- sibling(bob,S).

S = bob ;

S = albert ;

S = alice ;

S = bob ;

S = albert ;

S = alice ;

false.

Cut

Think:

• Any two people in the database are potentially good answers,
so we want to try all of them: can’t put “cut” in a rule after X
and/or Y is instantiated.

• i.e., can’t put ! anywhere inside the sibling(X,Y) rule.

• If 2 people share 2 parents, we only want to list them once as
the answer: want to put “cut” after both parent clauses.

⇒ need to redefine sibling: any two persons with at least one
common parent are siblings.

Cut

sibling(X,Y) :- person(X), person(Y), commonparent(X,Y).

person(X) :- male(X).

person(X) :- female(X).

commonparent(X,Y) :- parent(P,X), parent(P,Y), !.

?- sibling(bob,S).

S = bob ;

S = albert ;

S = alice ;

false.

The goal commonparent(bob,Y) is satisfied only once for each Y,
but all values of Y that satisfy sibling(X,Y) are found.

Cut and Negation

Finally, we don’t want X to be a sibling of X.

sibling(X,Y) :- \+(X=Y), person(X), person(Y),

commonparent(X,Y).

person(X) :- male(X).

person(X) :- female(X).

commonparent(X,Y) :- parent(P,X), parent(P,Y), !.

?- sibling(bob,S).

false.

What went wrong?

Cut and Negation

Finally, we don’t want X to be a sibling of X.

sibling(X,Y) :- \+(X=Y), person(X), person(Y),

commonparent(X,Y).

person(X) :- male(X).

person(X) :- female(X).

commonparent(X,Y) :- parent(P,X), parent(P,Y), !.

?- sibling(bob,S).

false.

What went wrong?

• \+(X=Y) should be placed after both X and Y are instantiated,
otherwise (bob=S) will always be true (why?), and hence
\+(bob=S) will always fail.

Cut and Negation

Solution:

sibling(X,Y) :- person(X), person(Y), \+(X=Y),

commonparent(X,Y).

person(X) :- male(X).

person(X) :- female(X).

commonparent(X,Y) :- parent(P,X), parent(P,Y), !.

?- sibling(bob,S).

S = albert ;

S = alice ;

false.

Cut: more examples

% sumAll(X,Y) holds iff Y = 1 + 2 + ... + X

sumAll(1, 1).

sumAll(N, Sum) :- N1 is N - 1,

sumAll(N1, S1),

Sum is S1 + N.

?- sumAll(2,S).

Cut: more examples

% sumAll(X,Y) holds iff Y = 1 + 2 + ... + X

sumAll(1, 1).

sumAll(N, Sum) :- N1 is N - 1,

sumAll(N1, S1),

Sum is S1 + N.

?- sumAll(2,S).

S = 3 ;

ERROR: Out of local stack

What went wrong?

Cut: more examples

% solution with Cut

sumAllC(1, 1) :- !.

sumAllC(N, Sum) :- N1 is N - 1,

sumAllC(N1, S1),

Sum is S1 + N.

?- sumAllC(2,S).

S = 3.

What does Cut do here?
Any other solutions (without using Cut)?

Cut: more examples

% deleteAll(X, L1, L2) holds iff L2 is the result of

% removing all instances of X from L1.

deleteAll(_,[],[]).

deleteAll(E,[E|Rest],L2) :- deleteAll(E, Rest, L2).

deleteAll(E,[X|Rest],[X|Rest2]) :-

deleteAll(E, Rest, Rest2).

?- deleteAll(a, [a,b,a], L).

L = [b] ;

L = [b, a] ;

L = [a, b] ;

L = [a, b, a] ;

false.

What is wrong?

Cut: more examples

% deleteAll(X, L1, L2) holds iff L2 is the result of

% removing all instances of X from L1.

deleteAll(_,[],[]).

deleteAll(E,[E|Rest],L2) :- !, deleteAll(E, Rest, L2).

deleteAll(E,[X|Rest],[X|Rest2]) :-

deleteAll(E, Rest, Rest2).

?- deleteAll(a, [a,b,a], L).

L = [b] ;

false.

Why did we put the Cut as the first subgoal in the second rule?
Can we put it after? What are we trying to achieve here?

Set manipulations

% subset(X,Y) holds iff X is a subset of Y.

% union(X,Y,U) holds iff U is union of X & Y.

% intersect(X,Y,I) holds iff I is intersection of X & Y.

Set manipulations

% subset(X,Y) holds iff X is a subset of Y.

subset([], Y).

subset([H|Rest], Y) :- member(H,Y), subset(Rest,Y).

Set manipulations

% union(X,Y,U) holds iff U is union of X & Y.

union([],Y,Y).

union([H|Rest],Y,U) :-

member(H,Y),

!,

union(Rest,Y,U).

union([H|Rest],Y,[H|Z]) :- union(Rest,Y,Z).

Set manipulations

% intersect(X,Y,I) holds iff I is intersection of X & Y.

intersect([],Y,[]).

intersect([H|Rest],Y,[H|Z]) :-

member(H,Y),

!,

intersect(Rest,Y,Z).

intersect([H|Rest],Y,Z) :- intersect(Rest,Y,Z).

